Maple Demo

General

Getting Help

There is an extensive online help feature. Help with individual commands can be accessed using the ?*commandname*. The options for a command are seen within that search. The results come up in a new window

```
> ?plot
```

Restart

It sometimes becomes necessary to clear the entire *Maple* memory. Use the restart command.

```
> restart;
```

Syntax

Maple uses lower case letters for the built-in functions and parentheses for the arguments of functions.

```
> sin(Pi/4);
```

Parentheses are also used for grouping symbols. Square brackets are used for lists and curly brackets are used for sets.
All statements must end with either a semicolon or colon. A colon will supress the output.

```
> 3+4;
  5*9:
```

Multiplication requires an asterisk, (with no exceptions.)

Functions

Functions are declared using the notation *f:=x->expression*

```
> f:=x->2*x-10;
> f(10);
```

Substitutions

Substitutions are done using the command subs

```
> g:=x^2;
> subs(x=4,g);
```

Numerical values

To get a numerical value, use evalf

```
> evalf(Pi);
  evalf(Pi,50);
```

Tables

Tables of values can be constructed from matrices but the built-in table command is not the same as in *Mathematica*.

```
> mytable:=array(1..4,1..2);
> for i from 1 to 4 do
  mytable[i,1]:=i:
  mytable[i,2]:=i^2:
  end do:
> print(mytable);
```

Graphics

Simple Plots

The standard plot command

```
> plot(x^2,x=-2..2,color=COLOR(HUE,0.5));
```

According to the help menu, the thickness can be set in values of 0, 1, 2, or 3. See plot[options]. Large numbers will work with unexpected results. Try 3, 15, and 16.

```
> plot(x^2,x=-2..2,color=COLOR(HUE,0.5),thickness=3);
```

Color

Colors can be declared either by the or *color=COLOR(HUE,#) or* *color=COLOR(RGB,#,#,#)* commands.Alternatively there are 25 built-in colors.

Combining Plots

By naming several plots, they can be combined using the display command. You must first load the with(plots) package. Also, The inital output of each plot must be supressed. background colors are not easily done.

```
> p1:=plot(x^2,x=-2..2,color=turquoise,thickness=5):
  p2:=plot(x^2+2,x=-2..2,color=plum,thickness=5):
  p3:=plot(x^2+4,x=-2..2,color=blue,thickness=5):
> with(plots):
> display(p1,p2,p3);
```

Notice that if you do not supress the output of the initial plot command, you get the data points.

```
> p1:=plot(x^2,x=-2..2,color=turquoise,thickness=5);
```

Surfaces

Surfaces can be plotted using the plot3d command

```
> plot3d(exp(-(x^2+y^2))*sin(x^2+y^2),x=-2..2,y=-2..2);
```

Maple's big advantage is is superior 3d rotation of graphics.

Parametric Functions

Parametric curves can be drawn through three different commands. *Maple* has inconsistent notation here.

```
> plot([cos(3*t)*cos(t),cos(3*t)*sin(t),t=0..2*Pi],color=blue,t
  hickness=3);
> spacecurve([cos(t)*cos(t),cos(t)*sin(t),sin(5*t),t=0..2*Pi],c
  olor=blue,thickness=3);
> plot3d([sin(u)*cos(v),sin(u)*sin(v),cos(v)],u=0..2*Pi,v=0..2*
  Pi,shading=zhue);
```

Maple also has a direct polar plot function that is not in *Mathematica.*

```
> polarplot(cos(3*theta),theta=0..2*Pi);
```

Animations

There are multiple ways of creating animations. I find the simplest is the animate command.

```
> animate( sin(x*t),x=-10..10,t=1..2,frames=10);
```

Calculus

Surface of Revolution

There is no specific surface of revolution command but the tubeplot command can help.

```
> tubeplot([t,0,0],t=-1..1,radius=t^2);
```

Derivatives

Maple can find the derivative of either a function or an expression. For a function,

```
> h:=x->x^5;
> D(h);
> D(h)(2);
```

For an expression, use

```
> m:=x^6;
  diff(m,x);
```

Integration

Integration can be done in either exact or numerical form

```
> int(sqrt(tan(x)),x=0..Pi/4);
  evalf(Int(sqrt(tan(x)),x=0..Pi/4));
```

Series

Infinite series are found using the series command

```
> myseries:=series(log(x),x=2,5);
```

To remove the "Big O" notation at the end, use the convert command

```
> convert(myseries,polynom);
```

Sum of a Series

To get the sum of a series, use the sum command

```
> mysum:=Sum(1/x,x=1..100)=sum(1/x,x=1..100);
>
```